HP-41CX Text File Output Solution

Introduction

The HP-41 provides the capability to create & store ASCII files in Extended Memory however, there is currently no way of directly exporting these ASCII files to DOS text files. 

The ability to create DOS text files from HP-41 file data is useful for getting information to other applications or to more easily share that information.

One possible solution is to use other output devices such as printers and RS232 adaptors. However, these solutions either require 'copying' a printer output or having actual real hardware neither of which are particularly easy or straight forward solutions. 

Another option is to use the virtual HP-IL DOS link application to write ASCII characters out directly to a DOS text file. However, for Extended Memory ASCII files this can only be performed via the Alpha register as a data output buffer for HP-IL. Unfortunately, because the HP-IL data stream does not know about file line & character pointers this technique is limited to the 24 character length of the Alpha register (actually 23 because the HP-IL 'OUT' functions require an initial 'dummy' (D) character placed in the Alpha register). As a result writing directly to a DOS text file using HP-IL from Extended Memory ASCII files limits file content to only 23 char long records creating files of the format:

QWERTYUIOPASDFGHJKLZXCV

QWERTYUIOPASDFGHJKLZXCV

QWERTYUIOPASDFGHJKLZXCV

...

If the records in Extended Memory are longer than 23 chars this information cannot be appended and is lost to the file. Furthermore, because the HP-41 ASCII files only know about records no other file structure e.g. fields, can be applied.

Solution

One solution is to copy the Extended Memory ASCII file into a more capable storage medium such as the ERAMCO storage unit, then use the ERAMCO RSU Database Software to create a file output program. The ERAMCO RSU Database Software has HP-IL output functions that can write up to 255 characters and provides programmable options as to whether CR/LF characters are appended to the character stream. Furthermore, the ERAMCO RSU storage unit enables fields to be created within records providing yet more flexibility in potential text file output requirements. These capabilities enable much larger records and text files to be created in the following format:

QWERTYUIOPASDFGHJKLZXCVBNM...255th char & CR/LF

QWERTYUIOPASDFGHJKLZXCVBNM...255th char & CR/LF

& CR/LF

Furthermore, the capability of the EREMCO RSU unit & software enables text file data to be imported into a suitably configured RSU text file creating a bi-directional data transfer capability!

System Configuration

The development of the following programs and ultimate solution was performed using the following components:

Virtual HP-41CX - For program development and HP-IL interface

Virtual HP-IL DOS Link - For DOS file input & output

Virtual  HP-8220B Printer - For fun!

For the complete process to work i.e. copying Extended Memory, Creating (output) DOS file and/or Importing DOS file the HP-41CX needs the following configuration:

2X HP82181A Extended Memory Modules

1X HP82160A HP-IL Module

1X HP82183A Extended I/O Module

1X 16K RAM for MLDL Module

1X ERAMCO ES-41 RSU Database Module

The ERAMCO Storage unit was initialized for 1 data block providing a total of 679 potential registers for ASCII or data files.

Programs

The following programs have been created to perform 3 functions:

1. Copy Extended (EM) Memory ASCII file (EM-RSU) to RSU file

2. Write (output) text files (OUTRSU)

3. Read (import) text files (INRSU)

EM-RSU

This program (EM-RSU) copies the specified Extended Memory ASCII file records to the specified RSU file directly as records. The program will copy records from the record number specified in X to file end i.e. max records.

The program first asks for the name of the Extended Memory file to copy and sets the record pointer accordingly (X) and then performs the same (name only) for the RSU file (pointer not required as data is appended to existing RSU file contents). The program then checks that the RSU file size is > the Extended Memory file size. If any of the file initialization operations fail (e.g. files don't exist or no data to copy etc.) the program exits.

If file initialization succeeds the program performs the 'Copy Record' (LBL 01) routine that loops through each Extended Memory file record and copies the record to the Alpha register and then appends the contents of the Alpha register as a new record in the RSU fie. 

The program then checks for any more characters in the record (F17 set) and if so executes the 'More Characters' (LBL 02) routine and appends them to the current RSU record until no more characters are available. The program then returns to the 'Copy Record' routine (LBL 01), if no more records are available the program performs the end routine and exits.

Inputs: X contains the record to start from (0 indexed)

Output: None (RSU file contains XMEM data)

Uses: Registers 00, 01, Stack & Alpha

Program Listing

LBL "EM-RSU"
;Copies EM ASCII file to RSU text file
 STO 01

;Stores start record
 RCLFLAG

;Get current flag status

 STO 00

;Store flags
 RCL 01

;Get start record
 CF 17

;Clear more chars flag

 CF 23

;Clear alpha input flag

 FIX 0


;Format display

 "EM FILE?"

;Name of EM file to copy prompt

 AON


;Enable alpha input

 PROMPT

;Display Alpha and wait for input

 AOFF


;Disable alpha input

 FC?C 23

;Is F23 clear? Will be set if alpha input was done

 GTO 10

;Yes failed alpha input so go to End routine

 SF 25


;Set system error flag

 SEEKPTA

;Sets record pointer to start of EM text file in Alpha

 FC? 25

;F25 is clear if no records

 GTO 10

;Yes no records to copy! so go to End routine

 "RSU FILE?"
;Name of RSU file for input prompt

 AON


;Enable alpha input

 PROMPT

;Display Alpha and wait for input

 AOFF


;Disable alpha input

 FC?C 23

;Is F23 clear? Will be set if alpha input was done

 GTO 10

;Yes failed alpha input so go to End routine

 SF 25


;Set system error flag

 RSTAF

;Sets record pointer to start of RSU text file in Alpha

 FC? 25

;F25 is clear if file not selected (found)

 GTO 10

;Yes no file so go to End routine

 CLA


;Clear Alpha for current file

 FLSIZE

;Get EM file register size to copy

 SZ?AF

;Get RSU file register size to copy

 X>Y?


;Is RSU file > EM file

 GTO 01

;Yes size ok go to Copy record routine

 GTO 10

;No size too small go to End routine

 LBL 01

;Copy record loop routine

 RCLPT

;Put EM file pointer in X

 INT


;Record pointer only

 VIEW X

;Display record pointer for progress indication

 SF 25


;Set system error flag

 GETREC

;Put EM file record into Alpha

 FC? 25

;F25 will be clear if no more records

 GTO 10

;Yes Go to end routine

 AREC

;No so append Alpha as new record in RSU file

 FS?C 17

;Are there more chars to get in record?

 XEQ 02

;Yes copy more chars

 GTO 01

;No go get any more records

 LBL 02

;Copy more chars routine

 GETREC

;Put <= 24 chars into Alpha

 ACHR

;Append Alpha chars to current RSU record

 FS?C 17

;Are there more chars to get in record?

 GTO 02

;Yes go to Copy more chars routine

 RTN


;No return to copy record routine

 LBL 10

;End routine

 RCL 00

;Get previous flags

 STOFLAG

;Restore flag status

 CLRG

;Clear stack

 CLST


;Clear registers

 CLA


;Clear Alpha
 CLD


;Clear display
 END
OUTRSU

This program (OUTRSU) copies the contents of the specified RSU file to the specified DOS file. The program first asks for the name of the RSU file to copy and sets the record pointer to the start of the file. The program then searches for the address of the virtual HP-IL DOS device (DOSLINK) makes the device the primary device and clears it. If the file initialization or device search operations fail (e.g. file doesn't exist or no DOSLINK device etc.) the program exits.

The program then gets the number of records in the file to copy and creates a control counter in the form rrr.eee. The program then performs a 'Send Record Data' routine that loops round for each record and sends the record data (with CR/LF appended) out to HP-IL. 

When all records have been processed the loop exits and an 'extra' CR/LF added to the file as an 'end of file' marker for the import program described later.

Program Prerequisites:

HP-IL loop is on

DOSLINK appropriately configured

Selected DOS Output/Input files exist

Input: RSU file name (in program)

Output: None

Uses: Register 01, Stack & Alpha

Program Listing

LBL "OUTRSU"
;Creates CR/LF delimited record DOS text file from RSU ASCII file

 CF 17

;F17 clear adds CR/LF to OUTREC

 CF 23

;Clear alpha input flag

 OUT FILE?

;Name of RSU file for output prompt

 AON


;Enable alpha input

 PROMPT

;Display Alpha and wait for input

 AOFF


;Disable alpha input

 FC?C 23

;Is F23 clear? Will be set if alpha input was done

 GTO 10

;Yes failed alpha input so go to End routine

 SF 25


;Set system error flag

 RSTAF

;Sets record pointer to start of RSU text file in Alpha

 FC? 25

;F25 is clear if file not selected (found)

 GTO 10

;Yes no file so go to End routine

 "DOSLINK"

;Device identifier (ID) in Alpha

 XROM 28,28
:FINDID returns address of DOSLINK device to X, else 0

 X=0?


;Device not found? (FINDID returns 0 if device not found)

 GTO 10

;Yes, device not found so go to End routine

 XROM 28,39
;SELECT selects the address of the found device to read from

 XROM 23,20
;CLRDEV clears the found device (device dependent)

 SF 25


;Set system error flag

 XROM 06,28
;IRCPT increment pointer to next record

 FC? 25

;F25 will be clear if no records (empty file)

 GTO 10

;Yes, so go to End routine (no records in file!)

 XROM 06,35
;NREC? get number of file records

 1000


;Create control counter rrr.nnn

 /

 1

 +

 STO 01

;Store control counter

 LBL 01

;Send record data loop

 XROM 06,45
;RRCPT Recalls record pointer value

 VIEW X

;Display pointer value for progress indication

 XROM 06,39
;OUTREC Record data to HP-IL & CR/LF, auto inc rec pointer

 ISG 01

;Increment control counter and skip if >

 GTO 01

;Go to Send record data loop

 "D"


;Dummy char in Alpha (ignored by OUTACL)

 XROM 23,36
;OUTACL Add extra CR/LF to be used as 'end of file' marker

 LBL 10

;End clear up routine

 XROM 23,20
;CLRDEV release & save file

 CLRG

;Clear registers

 CLST


;Clear stack

 CLA 


;Clear Alpha

 CLD


;Clear display

 END
Note: If the DOS file is to be used by another application it may be required to manually delete the 'extra' CR/LF' from the file.

INRSU

This program imports a DOS text file into a RSU text file. The program first asks for the name of the RSU file to copy and sets the record pointer accordingly. The program then searches for the address of the virtual HP-IL DOS device (DOSLINK) makes the device the primary device and clears it. If the file initialization or device search operations fail (e.g. file doesn't exist or no DOSLINK device etc.) the program exits.

The program then performs a 'Receive Data' routine (LBL 01), reading HP-IL until CR/LF encountered and checks Alpha register and appends content to RSU file as a record. The program then checks if more characters are left to get from the record (F17 set) and if so executes the 'More chars record' routine until no more characters (F17 clear). When all records have been processed the loop exits.

Program Prerequisites:

HP-IL loop is on

DOSLINK appropriately configured

Selected DOS Output/Input files exist

Input: RSU file name (in program)

Output: None

Uses: Stack & Alpha

Program Listing

LBL "INRSU"

;Reads CR/LF delimited record text file into RSU file

 CF 17

;F17 set more chars
 CF 23

;Clear alpha input flag

 RSU FILE?

;Name of RSU file for input prompt

 AON


;Enable alpha input

 PROMPT

;Display Alpha and wait for input

 AOFF


;Disable alpha input

 FC?C 23

;Is F23 clear? Will be set if alpha input was done

 GTO 10

;Yes so go to End routine

 SF 25


;Set system error flag

 RSTAF

;Sets record pointer to start of RSU text file in Alpha

 FC? 25

;Is file not selected (found)?

 GTO 10

;Yes no file so go to End routine

 "DOSLINK"

;Device identifier (ID) in Alpha

 XROM 28,28
:FINDID returns address of DOSLINK device to X, else 0

 X=0?


;Device not found? (FINDID returns 0 if device not found)

 GTO 10

;Yes, device not found so go to End routine

 XROM 28,39
;SELECT selects the address of the found device to read from

 XROM 23,20
;CLRDEV clears the found device (device dependent)

 LBL 01

;Receive data routine

 XROM 23,27
;INACL receives data to Alpha until CR/LF

 ATOX

;Clear 'D' from Alpha

 ALENG

;Alpha will now be clear if end of file

 X=0?


;No chars in Alpha?
 GTO 10

;Yes so go to end

 XROM 06,04
;AREC append alpha as new record to RSU file

 FS? 17

;F17 set if more chars in record

 XEQ 02

;Yes so execute More record chars routine

 XROM 06,45
;RRCPT Recalls record pointer value

 VIEW X

;Display pointer value for progress indication

 GTO 01

;Go to Receive data routine

 LBL 02

;More record chars routine

 XROM 23,27
;INACL receives data to Alpha until CR/LF

 ATOX

;Clear 'D' from Alpha

 XROM 06,01
;ACHR appends contents of Alpha to current record

 FC? 17

;CR/LF encountered

 RTN


;Yes so get next record

 GTO 02

;No get more chars

 LBL 10

;End clear up routine

 XROM 23,20
;CLRDEV release & save file

 CLRG

;Clear registers

 CLST


;Clear stack

 CLA 


;Clear Alpha

 CLD


;Clear display

 END
Potential Future Development

The main problem of not being able to write Extended Memory ASCII files (>24 char records) out directly to text files seems to be that no method of knowing the location of the text file pointer is (line no. & char pos). So the following enhanced DOSLINK features are proposed:

1. Enhancing the virtual HP-IL DOS Link to include device dependant functions for reading & writing text file line & character positions

2. Enhancing the virtual HP-IL DOS Link to include device dependant functions that enable text files to be created, selected, deleted (and moved/copied) in DOS paths

3. Enhancing the virtual HP-IL DOS Link to include say a 2k buffer to provide printer like capability and/or character accumulation functions

Enhancement 1 will enable additional character to be appended to the current record (line) of the text file providing > 23 character long records. 

Enhancement 2 will provide full DOS file management capability for the programmability of DOS file management.

Enhancement 3 will enable text files to be created under the control of printer functions? with the possibility of inserting field delimiting characters (e.g. ',' & TAB etc.) during character accumulation operations?

Additionally an extra file type (*.txt) could be added to the file selection drop down boxes on the DOSLINK interface?

